Points: Pie Graph

Query generates a mini pie chart for each point in the data. This query is data dependent, change yields to values appropriate for your data..



```
#Pie Graph of Crop Yield at each point
# yield values are converted to lbs and relative
#values are used to apportion the circle
numofcrops = 4          PieRadius = 10
array PartSize[4],Value[4],red[4],green[4],blue[4]

Value[1] = yield.wheat * .028
red[1] = 25 green[1] = 228 blue[1] = 155
Value[2] = yield.oats * .015
red[2] = 255 green[2] = 128 blue[2] = 45
Value[3] = yield.haydry
red[3] = 255 green[3] = 255 blue[3] = 25 #
Value[4] = yield.haywet
red[4] = 255 green[4] = 0 blue[4] = 0 # red

#find sum of values
SumValue =0
   for i = 1 to numofcrops step 1 {
    SumValue = Value[i] + SumValue
   }
#calculate partsize for each crop
  if (SumValue != 0) {
      for i = 1 to numofcrops step 1 {
      PartSize[i]= Value[i] *360 / SumValue
      }
   }
#draw pie piece fill areas
  LineStyleDropAnchor(0)  # center of symbol
  present = 0
  for crop = 1 to numofcrops step 1  {
       next = present + PartSize[crop]
       LineStyleSetColor(red[crop],green[crop],blue[crop])
        if (PartSize[crop] >0) {
           for angle = present to next -1 step 1 {
           LineStyleLineTo(angle,PieRadius)
             LineStyleMoveToAnchor(0)
           } present = next
         }
  }
 LineStyleSetColor(0,0,0)   #draw the edge lines in black
 angle = 0
 for crop = 1 to numofcrops step 1 { angle = angle + PartSize[crop]
    LineStyleLineTo(angle,PieRadius)
     LineStyleMoveToAnchor(0)
}
LineStyleDrawCircle(PieRadius) #draw outer circle
```